Supporting Information

Probing slow timescale dynamics in proteins using methyl ^{1}H CEST

Tairan Yuwen[1], Rui Huang[1] and Lewis E. Kay[1,2]

[1]Departments of Molecular Genetics, Biochemistry and Chemistry, University of Toronto, Toronto, Ontario, Canada

[2]Hospital for Sick Children, Program in Molecular Structure and Function, Toronto, Ontario, Canada

Table of Contents

Supplementary Figures **3**

References **6**

Pulse Sequence Code (Bruker) **7**

CH_3 labeling scheme 7

CHD_2 labeling scheme 16



Figure S1. Comparison of S/N ratios for spectra recorded with the pulse schemes of Figure 1A ($^{13}CH_3$-labeling) and 1B ($^{13}CHD_2$-labeling) for G48A Fyn SH3, 25 °C. Panel (A): ^{1}H R_1 rates measured for $^{13}CH_3$- and $^{13}CHD_2$-labeled methyl groups. Panel (B): Relative S/N ratios (SNR) of cross peaks in spectra of $^{13}CH_3$- and $^{13}CHD_2$-labeled proteins recorded with inter-scan delays of 1 s and 2 s, respectively, T_{Ex} = 0 and identical measurement times. Differences in inter-scan delays reflect the different R_1 values in each case, as indicated in Panel (A). The resulting S/N values have been corrected for the differences in concentrations of Fyn SH3 samples estimated from relative peak intensities in backbone ^{1}H–^{15}N HSQC datasets, as described previously (Rennella et al. 2015). Note that the predicted S/N ratios for $^{13}CH_3$- *vs* $^{13}CHD_2$- are $3/\sqrt{2}$ and $3/(2\sqrt{2})$ for Ile/Met and Leu/Val sites, respectively, neglecting relaxation. The slightly larger ratios reflect the increased sensitivity that derives from the larger ^{1}H R_1 rates in the case of proteins labeled as $^{13}CH_3$ (Rennella et al. 2015). Smaller differences in S/N values are obtained in difference CEST profiles, as described in the text.



Figure S2. Panel (A): Schematic defining linewidth, $\Delta\nu$, for anti-phase CEST dips obtained in difference CEST profiles. The width is taken as the difference in positions where the dip intensity is at half maximal value. Histograms of $\Delta\nu$ values of major state dips are shown for L99A T4L (Panel B, 9 °C) and $\alpha_7\alpha_7$ (Panel C, 50 °C).



Figure S3. Intensity of minor state dips in difference CEST profiles, I^{CEST}, obtained from simulations assuming a two-state exchange process with (p_E, k_{ex}) = (2%, 200 s^{-1}), T_{Ex} = 400 ms, as a function of relaxation delay between scans (d_1). It is assumed that there is no overlap between CEST and ground state dips, or CEST dips arising from distinct ^{13}C spin states, and further, that a weak B_1 field is used such that $B_1 > k_{ex}$ so that CEST dip sizes are maximal. Panels (A) and (B) are simulations using longitudinal relaxation parameters that are similar to those expected for $^{13}CH_3$- and $^{13}CHD_2$-labeled methyl groups, respectively. Note that the intensity of CEST dips decreases as ^{13}C R_1 rates increase due to a mixing of ^1H CEST multiplet components (Yuwen et al. 2017). Simulations were performed using the Bloch–McConnell equations, neglecting cross relaxation (see Eq [S16] of reference (Yuwen et al. 2017)).

References

E. Rennella, R. Huang, A. Velyvis, and L. E. Kay. $^{13}CHD_2$–CEST NMR spectroscopy provides an avenue for studies of conformational exchange in high molecular weight proteins. *J. Biomol. NMR*, 63(2):187–199, 2015. doi: 10.1007/s10858-015-9974-z.

T. Yuwen, A. Sekhar, and L. E. Kay. Separating dipolar and chemical exchange magnetization transfer processes in ^{1}H-CEST. *Angew. Chem. Int. Ed.*, 56(22):6122–6125, 2017. doi: 10.1002/anie.201610759.

```
/* 13CH3_1H_CEST_lek_800_cpz

  This pulse sequence will allow one to perform the following experiment:

  2D 1H/13C to measure exchange via 13CH3 CEST

  Assumes that sample is specifically 13C labeled

    1H: O1 on methyl 1H resonances (around 0.8 ppm)
        pwh = p1 1H pw90 @  power level pl1 highest power (13W)

    13C: O2 centre at 20 ppm
        pwc = p2 13C pw90 @ power level pl2 highest power (175W)

  To callibrate B1 field: set -Dcal_HB1
*/

;prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

/************************/
/*   Define phases     */
/************************/
#define zero ph=0.0
#define one ph=90.0
#define two ph=180.0
#define three ph=270.0

/**************************/
/*    Define pulses       */
/**************************/
define pulse dly_pg1     /* Messrle purge pulse */
  "dly_pg1=5m"
define pulse dly_pg2     /* Messrle purge pulse */
  "dly_pg2=dly_pg1/1.62"
define pulse pwh
  "pwh=p1"               /* 1H hard pulse at power level pl1 (tpwr) */
define pulse pwc
  "pwc=p2"               /* 13C hard pulse at power pl2 (dhpwr) */

#if defined(water_flg) || defined(buffer_flg)
  define pulse pw_sl1
    "pw_sl1=p14"         /* eburp1 pulse, ~7000 us */
#endif

/**************************/
/*    Define delays       */
/**************************/
define delay hscuba
  "hscuba=30m"
define delay taua        /* 1 / 4J(XH)        */
  "taua=d3"              /* 1s/(4*125) = 2 ms */

define delay taub        /* 1 / 4J(XH) or 1 / 8J(XH)  */
#if defined(s3_flg) && !defined(cal_HB1)
  "taub=1.03m"
#else
  "taub=2.0m"
#endif

#ifdef Met_flg
  "taua=taua*125/139"
  "taub=taub*125/139"
#endif

#if !defined(cal_HB1) && !defined(cal_HR1)
```

```
  define delay time_T1
    "time_T1=d17"         /* CEST duration - typically 100-500 ms */
#endif

"d11=30m"
"in0=inf1/2"             /* t1/2 increment */

/**************************************************************/
/* f1180: Start t1 at half dwell to get -90/180 phase correction */
/*        in F1 (15N) dim - set zgoptns -Df1180                  */
/**************************************************************/
#ifdef f1180
  "d0=larger((in0-pwc*4.0/PI-pwh*4.0)/2, 2e-7)"
#else
  "d0=larger((0.2u-pwc*4.0/PI-pwh*4.0)/2, 2e-7)"
#endif

/**********************************/
/*    Define other parameters    */
/**********************************/
#ifdef cal_HB1
  define list<delay> time_cal=<t1list_calB1>
#else
#ifdef cal_HR1
  define list<delay> time_cal=<t1list_calR1>
#else
  define list<frequency> H_offset=<$FQ1LIST>
#endif /*cal_HR1*/
#endif /*cal_HB1*/

/****************************************************/
/* Assign cnsts to check validity of parameter range */
/****************************************************/
#ifdef fsat
  "cnst10=plw10"         /* tsatpwr - set max at 0.00005W */
#endif

#ifdef mess_flg
  "cnst11=plw11"         /* tpwrmess - set max at 1W */
#endif

#if defined(water_flg) || defined(buffer_flg)
  "cnst14=spw14"         /* tpwrsl1 - set max at 0.001W */
#endif

/************************************************/
/* Set the weak H-amide B1 field for irradiation */
/************************************************/
"p15=1000000.0/(4.0*cnst16)"
"plw15=plw1*(p1)*(p1)/((p15)*(p15))"
"cnst15=plw15"

/*************************************/
/* Define macro for 3-9-19 WATERGATE */
/*************************************/
#define Watergate0 (center (pwh*0.231 ph26 d19*2 pwh*0.692 ph28 d19*2 pwh*1.462 ph26 d
19*2 pwh*1.462 ph26 d19*2 pwh*0.692 ph28 d19*2 pwh*0.231 ph26):f1 (pwc ph26 pwc ph28):
f2)
#define Watergate1 (center (pwh*0.231 ph26 d19*2 pwh*0.692 ph28 d19*2 pwh*1.462 ph26 d
19*2 pwh*1.462 ph26 d19*2 pwh*0.692 ph28 d19*2 pwh*0.231 ph26):f1 (pwc*2 ph26):f2)

/****************************/
/* Initialize loop counters */
/****************************/
"l2=0"
"l3=0"

;"acqt0=0"
;baseopt_echo
```

```
1 ze
/**************************************/
/* Check validity of parameter range  */
/**************************************/
#ifdef fsat
  if "cnst10 > 0.001"
  {
    2u
    print "error: presat power pl10 is too large"
    goto HaltAcqu
  }
#endif

#ifdef mess_flg
  if "cnst11 > 1.0"
  {
    2u
    print "error: tpwrmess pl11 too large"
    goto HaltAcqu
  }
#endif

  if "dly_pg1 > 10m"
  {
    2u
    print "error:dly_pg too long !! "
    goto HaltAcqu
  }

#if defined(water_flg) || defined(buffer_flg)
  if "cnst14 > 0.01"
  {
    2u
    print "error: power level for eburp1 pulse is too large"
    goto HaltAcqu
  }
#endif

  if "cnst15 > 0.01"
  {
    2u
    print "error: power level for 1H cest is too large"
    goto HaltAcqu
  }

  if "cnst16 > 250"
  {
    2u
    print "error: cnst16 - weak 1H B1 field strength in Hz must be <= 250 Hz !!!  "
    goto HaltAcqu
  }

  if "pwc > 20u"
  {
    2u
    print "error: pwc too large < 20 us"
    goto HaltAcqu
  }

#if !defined(cal_HB1) && !defined(cal_HR1)
  if "time_T1 > 0.5s"
  {
    2u
    print "error: time_T1 is too long"
    goto HaltAcqu
  }
#endif
```

```
2 d11
/****************************************/
/* Continue to check real time variables */
/****************************************/
#ifdef cal_HB1
  "DELTA = time_cal[l2]"
  if "DELTA > 500m" {
    2u
    print "error: time_cal (evolution time) during B1 calib is too long"
    goto HaltAcqu
  }
#endif

  20u fq=cnst1:f1        /* jump 1H carrier to water */

/******************/
/* Messerle purge */
/******************/
#ifdef mess_flg
  20u pl11:f1
  (dly_pg1 ph26):f1
  20u
  (dly_pg2 ph27):f1
  20u pl10:f1
#endif /*mess_flg*/

/*****************/
/* Presaturation */
/*****************/
#ifdef fsat
  4u pl10:f1
  d1 cw:f1 ph26
  4u do:f1
  2u pl1:f1
#ifdef fscuba
  hscuba
  (pwh ph26):f1
  (pwh*2 ph27):f1
  (pwh ph26):f1
  hscuba
#endif /*fscuba*/
#else
  2u pl1:f1
  d1
#endif /*fsat*/

  20u UNBLKGRAD

/***************************************/
/* Erase 13C equilibrium magnetization */
/***************************************/
  2u pl2:f2
  (pwc ph26):f2

  2u
  p50:gp0
  d16

/***************************************************/
/*  +/- phase cycle on 1H prior to CEST period    */
/***************************************************/
#ifdef Hsteady_flg
  20u fq=0:f1            /* jump 1H carrier back to methyl */

  (pwh ph26):f1
  (pwh*2 ph0):f1
  (pwh ph26):f1

  2u
```

```
  p50:gp0
  d16
#endif /*Hsteady_flg*/

/*****************************************/
/*   CEST portion on Hz magnetization   */
/*****************************************/
  2u pl15:f1
#ifdef cal_HB1
  "DELTA=time_cal[l2]"
  2u fq=cnst3:f1
  DELTA cw:f1 ph26
  2u do:f1
#else
#ifdef cal_HR1
  "DELTA=time_cal[l2]"
  DELTA
#else
  2u fq=H_offset:f1
  time_T1 cw:f1 ph26
  2u do:f1
#endif /*cal_HR1*/
#endif /*cal_HB1*/
  20u fq=cnst1:f1        /* jump 1H carrier to water */

/************************************************************/
/*  Destroy water and buffer magnetization at the beginning  */
/************************************************************/
#ifdef water_flg
  (pw_sl1:sp14 ph26):f1
#endif

#ifdef buffer_flg
  20u fq=cnst2:f1        /* jump 1H carrier to buffer position */
  (pw_sl1:sp14 ph26):f1
#endif

  2u
  p50:gp0
  d16

  20u fq=0:f1            /* jump 1H carrier back to methyl */
  2u pl1:f1 pl2:f2

/************************************************/
/* H->C transfer that preserves the spin state */
/************************************************/
  (pwh ph26):f1

#ifndef Hsteady_flg
  (pwc ph1):f2
#else
  (pwc ph11):f2
#endif

  2u
  p51:gp1
  d16

  "DELTA = taua - 2u - p51 - d16"
  DELTA

  (center (pwh ph26 pwh*2 ph27 pwh ph26):f1 (pwc ph26 pwc*2 ph27 pwc ph26):f2)

  DELTA

  2u
  p51:gp1
  d16
```

```
/*********************/
/*   13C labeling    */
/*********************/
#ifndef Hsteady_flg
  (pwc ph2):f2
#else
  (pwc ph12):f2
#endif

  d0

  (pwh ph26):f1
  (pwh*2 ph27):f1
  (pwh ph26):f1

  d0

/***************************************************/
/* C->H back transfer that preserves the spin state */
/***************************************************/
  (pwc ph3):f2

  2u
  p51:gp1
  d16

  "DELTA = taua - 2u - p51 - d16"
  DELTA

  (center (pwh ph26 pwh*2 ph27 pwh ph26):f1 (pwc ph26 pwc*2 ph27 pwc ph26):f2)

  DELTA

  2u
  p51:gp1
  d16

  if "l3%2 == 0" {
    (pwc ph4):f2
  }
  else {
    (pwc ph5):f2
  }

  (pwh ph27):f1
#ifndef Hsteady_flg
  (pwh ph6):f1
#else
  (pwh ph16):f1
#endif

/*******************************************************/
/*   IPAP/S3E for separating two spin states           */
/*   use -Dwtg_flg to achieve better water suppression  */
/*******************************************************/
#ifdef wtg_flg
  2u
  p52:gp2
  d16

  "DELTA = taub - pwc - 2u - p52 - d16 - d19*3.367"
  DELTA

#ifdef s3_flg
  Watergate1
#else
  if "l3%2 == 0" {
    Watergate0
```

```
  }
  else {
    Watergate1
  }
#endif /*s3_flg*/

#ifdef s3_flg
  "DELTA = taub - pwc - 2u - p52 - d16 - d19*3.367 + pwh*2.0/PI - 4u - de - pwc*2"
#else
  "DELTA = taub - pwc - 2u - p52 - d16 - d19*3.367 + pwh*2.0/PI - 4u - de"
#endif /*s3_flg*/
  DELTA

  2u
  p52:gp2
  d16
#else
  2u
  p53:gp3
  d16

  "DELTA = taub - pwc - 2u - p53 - d16"
  DELTA

#ifdef s3_flg
  (center (pwh ph26 pwh*2 ph27 pwh ph26):f1 (pwc*2 ph26):f2)
#else
  if "l3%2 == 0" {
    (center (pwh ph26 pwh*2 ph27 pwh ph26):f1 (pwc ph26 pwc ph28):f2)
  }
  else {
    (center (pwh ph26 pwh*2 ph27 pwh ph26):f1 (pwc ph26 pwc ph26):f2)
  }
#endif /*s3_flg*/

#ifdef s3_flg
  "DELTA = taub - pwc - 2u - p53 - d16 + pwh*2.0/PI - 4u - de - pwc*2"
#else
  "DELTA = taub - pwc - 2u - p53 - d16 + pwh*2.0/PI - 4u - de"
#endif /*s3_flg*/
  DELTA

  2u
  p53:gp3
  d16
#endif /*wtg_flg*/

  4u BLKGRAD

#ifdef s3_flg
  if "l3%2 == 0" {
    (pwc ph26 pwc ph26):f2
  }
  else {
    (pwc ph26 pwc ph28):f2
  }
#endif /*s3_flg*/

  if "l3%2 == 0" {
    go=2 ph30
  }
  else {
    go=2 ph31
  }
  d11 mc #0 to 2

  F3QF(calclc(l3,1))
#if defined(cal_HB1) || defined(cal_HR1)
  F2QF(calclc(l2,1))
```

```
#else
  F2QF(calclist(H_offset,1))
#endif
  F1PH(calph(ph1, -90) & calph(ph2, +90) & calph(ph11, -90) & calph(ph12, +90), calph(
ph1, +180) & calph(ph2, +180) & calph(ph11, +180) & calph(ph12, +180) & calph(ph30,+18
0) & calph(ph31, +180) & caldel(d0, +in0))

HaltAcqu, 1m
exit

ph0=2 1
ph1=1 3 3 1
ph2=0 2 2 0
ph3=0 0 0 0 2 2 2 2
ph4=1 1 1 1 3 3 3 3
ph5=3 3 3 3 1 1 1 1
ph6=3 1
ph11=1 1 3 3
ph12=0 0 2 2
ph16=3 3 1 1
ph26=0
ph27=1
ph28=2
ph29=3
ph30=0 2 2 0 2 0 0 2
ph31=1 3 3 1 3 1 1 3

;d1 : repetition delay
;d3 : taua ~1/4J(CH) 2.00 ms
;d11 : delay for disk i/o, 30ms
;d16 : gradient recovery delay, 200us
;d17 : time_T1 <= 500 ms
;d19 : delay for binomial water suppression
;      d19 = (1/(4*|cnst1|)), cnst1 = distance between methyl and water (in Hz)
;pl1 : tpwr - power level for pwh
;pl2 : dhpwr - power level for hard 13C pulse pwc
;pl15 : power level for 1H cest
;spw14 : power level for eburp1 pulse
;spnam14: eburp1 pulse on water
;p1 : pwh
;p2 : pwc
;p14 : eburp1 pulse width, typically 7000u
;p15 : pw of 1H for cest element
;p50: gradient pulse 50                                [1000 usec]
;p51: gradient pulse 51                                [500 usec]
;p52: gradient pulse 52                                [800 usec]
;p53: gradient pulse 53                                [500 usec]
;cnst1 : water(Hz) - methyl(Hz)
;cnst2 : buffer(Hz) - methyl(Hz)
;cnst3 : offset (Hz) for B1 calibration from o1
;cnst14 : power level in w for 1H selective pulse
;cnst15 : power level in w for weak 1H B1 cest field
;cnst16 : 1H B1 field to use for CEST, typically 15-50 Hz
;l2 : pointer to the delay value for calibration of B1
;l3 : pointer to switch between IP and AP
;inf1 : 1/SW(X) = 2*DW(X)
;in0 : 1/(2*SW(x))=DW(X)
;nd0 : 2
;ns : 2*n
;FnMODE : States, TPPI or States-TPPI

;for z-only gradients:
;gpz0: 50%
;gpz1: 30%
;gpz2: 80%
;gpz3: 40%

;use gradient files:
;gpnam0: SMSQ10.32
```

```
;gpnam1: SMSQ10.32
;gpnam2: SMSQ10.32
;gpnam3: SMSQ10.32

;zgoptns : Df1180, Dfsat, Dmess_flg, Dfscuba, Dwater_flg, Dbuffer_flg, Dcal_HB1, Dwtg_
flg, Ds3_flg, DHsteady_flg, Dcal_HR1, DMet_flg
```

```
/* 13CHD2_1H_CEST_lek_800_cpz

  This pulse sequence will allow one to perform the following experiment:

  2D 1H/13C to measure exchange via 13CHD2 CEST

  Option for 2H decoupling (used for 13CD2)

  Assumes that sample is specifically 13C labeled

    1H: O1 on methyl 1H resonances (around 0.8 ppm)
        pwh = p1 1H pw90 @  power level pl1 highest power (13W)

    13C: O2 centre at 20 ppm
        pwc = p2 13C pw90 @ power level pl2 highest power (175W)

  To callibrate B1 field: set -Dcal_HB1

  Used with N15cest_make_fq_list.c (ccode) which makes the frequency list
  Used with N15cest_cal_B1.c (ccode)
  Used with move_lek_cest_fq which moves the freq list to f1
  Used with move_lek_cest_B1 which moves the list of times to vd list

  Het-CP back transfer total time: 51.8*p_d*4*2 = 230.2 * pw_cp , so set pw_cp to 34.5
0 us

  RUN THIS EXPT WITH lock = H2O_D2O NOT D2O even if you are in a D2O sample - or you m
ay lose lock
  Use lfilter = 500
  did further tests and isolated the problem to the purge pulse/gradient pair after th
e 1H heating
  must use gramp not grad for some reason and then works fine for H2O_D2O or H2O solve
nt
*/

;prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

/***********************/
/*   Define phases     */
/***********************/
#define zero ph=0.0
#define one ph=90.0
#define two ph=180.0
#define three ph=270.0

/**************************/
/*    Define Pulses       */
/**************************/
define pulse dly_pg1     /* Messerle purge pulse */
  "dly_pg1=5m"
define pulse dly_pg2     /* Messerle purge pulse */
  "dly_pg2=dly_pg1/1.62"
define pulse pwh
  "pwh=p1"               /* 1H hard pulse at power level p1 (tpwr) */
define pulse pwc
  "pwc=p2"               /* 13C pulse at power level pl2 (dhpwr) */
define pulse pw_dip
  "pw_dip=p14"
define pulse p_d         /* pulse for het cp transfer back to 1H */
  "p_d=pw_dip*5.0/9.0"
define pulse pwd
  "pwd=p4"               /* 2H pulse at power pl4 */

/**************************/
/*    Define delays       */
```

```
/**************************/
define delay hscuba      /* length of 1/2 scuba delay */
  "hscuba=25m"
define delay taua        /* 1 / 4J(XH)                */
  "taua=d3"              /* 1s/(4*125) = 2.0 ms */

define delay taub        /* 1 / 4J(XH) or 1 / 8J(XH)  */
#if defined(s3_flg) && !defined(cal_HB1)
  "taub=1.03m"
#else
  "taub=2.0m"
#endif

#ifdef Met_flg
  "taua=taua*125/139"
  "taub=taub*125/139"
#endif

#if !defined(cal_HB1) && !defined(cal_HR1)
  define delay time_T1
    "time_T1=d17"        /* CEST duration - typically 100-500 ms */
#endif

"in0=inf1/2"
"d11=30m"
"d31=(2590.0/90.0)*p14*8"

/*************************************************************/
/* f1180: Start t1 at half dwell to get -90/180 phase correction */
/*        in F1 (15N) dim - set zgoptns -Df1180                */
/*************************************************************/
#ifdef f1180
  "d0=(in0)/2"
#else
  "d0=0.2u"
#endif

/********************************/
/*    Define other parameters    */
/********************************/
#ifdef cal_HB1
  define list<delay> time_cal=<t1list_calB1>
#else
#ifdef cal_HR1
  define list<delay> time_cal=<t1list_calR1>
#else
  define list<frequency> H_offset=<$FQ1LIST>
#endif
#endif

/****************************************************/
/* Assign cnsts to check validity of parameter range */
/****************************************************/
#ifdef fsat
  "cnst10=plw10"         /* tsatpwr - set max at 0.00005W */
#endif

#ifdef mess_flg
  "cnst11=plw11"         /* tpwrmess pl11 - set max at 1.0W */
#endif

"cnst14=plw14"           /* power level for het-cp back transfer (1H) */
"cnst22=plw22"           /* power level for het-cp back transfer (13C) */

#ifdef Ddec
  "cnst4=plw4"           /* dpwr3 pl4 - set max at 10.5W */
  "cnst41=plw41"         /* dpwr3D pl41 - set max at 1.5W */
#endif
```

```
/*************************************************/
/* Set the weak H-amide B1 field for irradiation */
/*************************************************/
"p15=1000000.0/(4.0*cnst16)"
"plw15=plw1*(p1)*(p1)/((p15)*(p15))"
"cnst15=plw15"

/***************************/
/* Initialize loop counters */
/***************************/
"l2=0"
"l3=0"

;"acqt0=0"
;baseopt_echo

1 ze
/**************************************/
/* Check validity of parameter range  */
/**************************************/
#ifdef fsat
  if "cnst10 > 0.00005"
  {
    2u
    print "error: tpwrmess pl10 too large"
    goto HaltAcqu
  }
#endif

#ifdef mess_flg
  if "cnst11 > 1.0"
  {
    2u
    print "error: tpwrmess pl11 too large"
    goto HaltAcqu
  }
#endif

  if "dly_pg1 > 10m"
  {
    2u
    print "error:dly_pg too long !! "
    goto HaltAcqu
  }

  if "cnst15 > 0.01"
  {
    2u
    print "error: power level for 1H cest is too large"
    goto HaltAcqu
  }

  if "cnst16 > 250"
  {
    2u
    print "error: cnst16 - weak 1H B1 field strength in Hz must be <= 250 Hz !!!  "
    goto HaltAcqu
  }

  if "cnst14 > 2.0"
  {
    2u
    print "error: 1H power for het cp back transfer pl14 too large < 2 W"
    goto HaltAcqu
  }

  if "cnst22 > 40.0"
  {
    2u
```

```
    print "error: 13C power for hetcp back transfer is too large < 40W"
    goto HaltAcqu
  }

  if "p_d > 35u"
  {
    2u
    print "error: p_d is too long; must be < 35u"
    goto HaltAcqu
  }

#if !defined(cal_HB1) && !defined(cal_HR1)
  if "time_T1 > 0.5s"
  {
    2u
    print "error: time_T1 is too long"
    goto HaltAcqu
  }
#endif

#ifdef Ddec
  if "cnst4 > 13.5"
  {
    2u
    print "error: dpwr3 pl4 too large"
    goto HaltAcqu
  }

  if "cnst41 > 3.0"
  {
    2u
    print "error: dpwr3D pl41 too large"
    goto HaltAcqu
  }

  if "pwd > 150u" {
    2u
    print "error: pwd is too large < 150u"
    goto HaltAcqu
  }

  if "pwd < 100u" {
    2u
    print "error: pwd is too small > 100u"
    goto HaltAcqu
  }

; d11 LOCKDEC_ON         /* Not required for AvanceIII-HD */
  50u LOCKH_ON
  d11 H2_PULSE
  2u pl41:f4
#endif

2 d11
/******************************************/
/* Continue to check real time variables */
/******************************************/
  20u fq=cnst1:f1        /* jump 1H carrier to water */
  20u fq=0:f2
  2u pl1:f1 pl2:f2 pl3:f3

#ifdef Ddec              /* D decoupling */
  d11 H2_LOCK            /* put lock channel in lock mode */
  6m LOCKH_OFF           /* turn off lock hold */
#endif /*Ddec*/

/*****************/
/* Messerle purge */
/*****************/
```

```
#ifdef mess_flg
  20u pl11:f1
  (dly_pg1 ph26):f1
  20u
  (dly_pg2 ph27):f1
  20u pl10:f1
#endif /*mess_flg*/

/****************/
/* Presaturation */
/****************/
#ifdef fsat
  4u pl10:f1
  d1 cw:f1 ph26
  4u do:f1
  2u pl1:f1
#ifdef fscuba
  hscuba
  (pwh ph26):f1
  (pwh*2 ph27):f1
  (pwh ph26):f1
  hscuba
#endif /*fscuba*/
#else
  2u pl1:f1
  d1
#endif /*fsat*/

#ifdef Ddec
  50u LOCKH_ON
  15u H2_PULSE
#endif

/****************************************/
/* Erase 13C equilibrium magnetization */
/****************************************/
  (pwc ph26):f2

#ifdef Ddec
  20u UNBLKGRAMP
#else
  20u UNBLKGRAD
#endif

  2u
  p50:gp0
  d16

/*****************************************************/
/*  +/- phase cycle on 1H prior to CEST period    */
/*****************************************************/
#ifdef Hsteady_flg
  20u fq=0:f1            /* jump 1H carrier back to methyls */

  (pwh ph26):f1
  (pwh*2 ph0):f1
  (pwh ph26):f1

  2u
  p50:gp0
  d16
#endif /*Hsteady_flg*/

/****************************************/
/*   CEST portion on Hz magnetization   */
/****************************************/
  2u pl15:f1
#ifdef cal_HB1
  "DELTA=time_cal[l2]"
```

```
  2u fq=cnst3:f1
  DELTA cw:f1 ph26
  2u do:f1
#else
#ifdef cal_HR1
  "DELTA=time_cal[l2]"
  DELTA
#else
  2u fq=H_offset:f1
  time_T1 cw:f1 ph26
  2u do:f1
#endif /*cal_HR1*/
#endif /*cal_HB1*/
  20u fq=0:f1             /* jump 1H carrier back to methyls */
  2u pl1:f1
  2u rpp4 rpp5            /* reset phase tables for het-cp */

  2u
  p50:gp0
  d16

/***********************************/
/*   H->C transfer with ST2-PT    */
/***********************************/
#ifdef Ddec
  2u pl4:f4
  (pwd ph27):f4
  2u pl41:f4
  (2u cpds4 ph26):f4
#endif

  (pwc ph26):f2

#ifdef Ddec
  taua
#else
  2u
  p51:gp1
  d16

  "DELTA = taua - 2u - p51 - d16"
  DELTA
#endif

  (center (pwh*2 ph26):f1 (pwc*2 ph26):f2)

#ifdef Ddec
  taua
#else
  DELTA

  2u
  p51:gp1
  d16
#endif

  (pwc ph27):f2
  (pwh ph26):f1

#ifdef Ddec
  taua
#else
  2u
  p52:gp2
  d16

  "DELTA = taua - 2u - p52 - d16"
  DELTA
#endif
```

```
  (center (pwh*2 ph26):f1 (pwc*2 ph26):f2)

#ifdef Ddec
  taua
#else
  DELTA

  2u
  p52:gp2
  d16
#endif

  (pwh ph27):f1

#ifdef Ddec
  2u do:f4
  2u pl4:f4
  (pwd ph29):f4
#endif

  2u
  p53:gp3
  d16

#ifdef Ddec
  2u pl4:f4
  (pwd ph27):f4
  2u pl41:f4
  (2u cpds4 ph26):f4
#endif

/*********************/
/*   13C labeling    */
/*********************/
#ifndef Hsteady_flg
  (pwc ph1):f2
#else
  (pwc ph11):f2
#endif

  d0
  (pwh ph26):f1
  (pwh*2 ph27):f1
  (pwh ph26):f1
  d0

#ifdef Ddec
  2u do:f4
  2u pl4:f4
  (pwd ph29):f4
#endif

/***********************************************/
/*   Encoding coherence selection gradient    */
/***********************************************/
  2u
  p54:gp4*EA
  d16

  (pwc*2 ph3):f2

  2u
  p54:gp4*EA*-1.0
  d16

#ifdef Ddec
  "DELTA = 2u + pwd + 2u"
  DELTA
```

```
#endif

  (pwh ph26 pwh*2.0 ph27 pwh ph26):f1

/****************************************************/
/*   Planar mixing C->H back transfer with het-cp   */
/****************************************************/
  (ralign (pwh ph27):f1 (pwc ph2):f2)

  1u pl14:f1
  1u pl22:f2

6 (center (p_d*6.4 ph4):f1 (p_d*6.4 ph4):f2)
  (center (p_d*8.2 ph5):f1 (p_d*8.2 ph5):f2)
  (center (p_d*5.8 ph4):f1 (p_d*5.8 ph4):f2)
  (center (p_d*5.7 ph5):f1 (p_d*5.7 ph5):f2)
  (center (p_d*0.6 ph4):f1 (p_d*0.6 ph4):f2)
  (center (p_d*4.9 ph5):f1 (p_d*4.9 ph5):f2)
  (center (p_d*7.5 ph4):f1 (p_d*7.5 ph4):f2)
  (center (p_d*5.3 ph5):f1 (p_d*5.3 ph5 ipp5):f2)
  (center (p_d*7.4 ph4):f1 (p_d*7.4 ph4 ipp4):f2)
lo to 6 times 8

  1u pl1:f1
  1u pl2:f2

  if "l3 % 2 == 0" {
    (lalign (pwh ph27):f1 (pwc ph27):f2)
  }
  else {
    (lalign (pwh ph27):f1 (pwc ph29):f2)
  }

/**********************************************************/
/*   IPAP/S3E and decoding coherence selection gradient   */
/**********************************************************/
  taub

#ifdef s3_flg
  (center (pwh*2 ph26):f1 (pwc*2 ph26):f2)
#else
  if "l3%2 == 0" {
    (center (pwh*2 ph26):f1 (pwc ph26 pwc ph28):f2)
  }
  else {
    (center (pwh*2 ph26):f1 (pwc*2 ph26):f2)
  }
#endif

#ifdef s3_flg
  "DELTA = taub - 2u - p55 - d16 - 4u - de - pwc*2"
#else
  "DELTA = taub - 2u - p55 - d16 - 4u - de"
#endif
  DELTA

  2u
  p55:gp5
  d16

#ifdef Ddec
  4u BLKGRAMP
#else
  4u BLKGRAD
#endif

#ifdef s3_flg
  if "l3%2 == 0" {
    (pwc ph26 pwc ph26):f2
```

```
  }
  else {
    (pwc ph26 pwc ph28):f2
  }
#endif

  if "l3%2 == 0" {
    go=2 ph30
  }
  else {
    go=2 ph31
  }

  d11 mc #0 to 2

  F3QF(calclc(l3,1))
#if defined(cal_HB1) || defined(cal_HR1)
  F2QF(calclc(l2,1))
#else
  F2QF(calclist(H_offset,1))
#endif /*cal_HB1,cal_HR1*/
  F1EA(calgrad(EA) & calph(ph2,+180), calph(ph1,+180) & calph(ph11,+180) & calph(ph30,
+180) & calph(ph31,+180) & caldel(d0,+in0))

#ifdef Ddec
  d11 H2_LOCK
  d11 LOCKH_OFF
; d11 LOCKDEC_OFF        /* use statement for earlier hardware */
#endif

HaltAcqu, 1m
exit

ph0=2 1
ph1=0 2 2 0
ph2=1
ph3=0 0 0 0 1 1 1 1 2 2 2 2 3 3 3 3
ph4=0 2 2 0
ph5=2 0 0 2
ph11=0 0 2 2
ph26=0
ph27=1
ph28=2
ph29=3
ph30=0 2 2 0 2 0 0 2
ph31=1 3 3 1 3 1 1 3

;pl1 : tpwr - power level for pwh
;pl2 : dhpwr - power level for 13C pulse pwc (p2)
;pl10 : tsatpwr - power level for presat
;pl11 : tpwrmess - power level for Messerle purge
;pl14 : 1H power level for het cp back transfer
;pl15 : power level for 1H cest
;pl22 : power level for 13C het cp back transfer
;p1 : pwh
;p2 : pwc
;p4 : 2H high power pulse
;p14 : 1H pw90 for hetcp back transfer ~ 34.50 us
;      transfer time = 230.2*p14 = 1/JCH
;p50 : gradient pulse 50                              [1000 usec]
;p51 : gradient pulse 51                              [500 usec]
;p52 : gradient pulse 52                              [500 usec]
;p53 : gradient pulse 53                              [1000 usec]
;p54 : gradient pulse 54                              [512 usec]
;p55 : gradient pulse 55                              [256 usec]
;pcpd4 : 2H pulse width for 2H decoupling
;d1 : Repetition delay D1
;d3 : taua ~1/(4*JCH)  ~2ms
;d11 : delay for disk i/o, 30ms
```

```
;d16 : gradient recovery delay, 200us
;d17 : time_T1 <= 500 ms
;d31 : dipsi2 het-cp back transfer duration (~1/JCH = 8 ms)
;cpd4 : 2H decoupling during t1
;cpdprg4 : 2H decoupling during t1
;cnst1 : water(Hz) - methyl(Hz)
;cnst3 : offset (Hz) for B1 calibration from o1
;cnst4 : power level for 2H high power pulses
;cnst14 : 1H power in w for hetCP
;cnst15 : power in w for 1H CEST B1
;cnst16 : 1H B1 field to use for CEST, typically 15-50 Hz
;cnst22 : power for 13C for het cp
;cnst41 : power level for 2H decoupling
;l2 : pointer to the delay value for calibration of B1
;l3 : pointer to switch between IP and AP
;infl : 1/SW(X) = 2*DW(X)
;in0 : 1/(2*SW(x))=DW(X)
;nd0 : 2
;ns : 2*n
;FnMODE : echo-antiecho

;use gradient ratio:    gp 4 : gp 5
;                         90 :   90

;for z-only gradients:
;gpz0: 40%
;gpz1: 30%
;gpz2: 40%
;gpz3: -50%
;gpz4: 90%
;gpz5: 90%

;use gradient files:
;gpnam0: SMSQ10.32
;gpnam1: SMSQ10.32
;gpnam2: SMSQ10.32
;gpnam3: SMSQ10.32
;gpnam4: SMSQ10.32
;gpnam5: SMSQ10.32

;zgoptns : Df1180, Dfsat, Dmess_flg, Dfscuba, Dcal_HB1, DDdec, DHsteady_flg, Dcal_HR1,
 DMet_flg
```